











Figure 1. Firewall structure of the Lanbooster.












Figure 2. The flowchart of a Firewall.

1. Introduction of the IP Filter

Your Lanbooster is equipped with a powerful IP filter that can be programmed by professional users to function as a firewall. The IP filters are implemented at three interfaces of the ISDN router, namely, the LAN interface, the ISP (dial-up) interface as well as the Dial-in (remote access) interface. Please refer to Figure 1. for the corresponding interfaces.

There are two sets (or profiles) of filter rules that can be swapped via a Telnet command to make one of them become active and the other inactive. The packets passed through or blocked by the filter can be logged for further analysis. The log function follows the principles specified both implicitly in the filter rules and explicitly of a Telnet command.

The in/out naming convention used in the IP filter is also depicted in Figure 1. Every packet intends to pass the filter will be processed as shown in Figure 2. The IP Accounting function (via the 'count' action in the filter rule) causes the packet to be included in the accounting statistics kept by the filter. This is followed by two cache matches allowing packets which match these to flow straight through, rather than going through the filter rule list. The 'keep frags' and 'keep state' filter rule options can record historical information on fragmented packets and information about the flow of a communication session respectively for this purpose.
The IP Filtering goes through a tree-like filter list, initially in the trunk (group 0) and branches whenever the packet matches a rule with the group 'head' option specifying a new group. There are three possible outcomes of the filtering---Pass, Block, and No-match. The filter will let go the packets with the no-match result.

2. The Filter Rules

2.1 Filter Rule Syntax

filter-rule = [ insert ] action in-out [ options ] [ tos ] [ ttl ]


      [ proto ] [ ip ] [ group ].
insert
= "@" decnumber .

action
= block | "pass" | log | "count" | skip .

in-out
= "in" | "out" .

options
= [ log ] [ "quick" ] [ "on" interface-name [ dup ] ] .

tos
= "tos" decnumber | "tos" hexnumber .

ttl
= "ttl" decnumber .

proto
= "proto" protocol .

ip
= srcdst [ flags ] [ with withopt ] [ icmp ] [ keep ] .

group
= [ "head" decnumber ] [ "group" decnumber ] .

block
= "block" [ icmp[return-code] | "return-rst" ] .

log
= "log" [ "body" ] [ "first" ].

interface-name
= "lan" | "isp" | "dialin" .

skip
= "skip" decnumber .

dup
= "dup-to" interface-name[":"ipaddr] .

protocol = "tcp/udp" | "udp" | "tcp" | "icmp" | decnumber .

srcdst
= "all" | fromto .

fromto
= "from" object "to" object .

icmp
= "return-icmp" | "return-icmp-as-dest" .

object
= addr [ port-comp | port-range ] .

addr
= "any" | nummask | host-name [ "mask" ipaddr | "mask" hexnumber ] .

port-comp = "port" compare port-num .

port-range = "port" port-num range port-num .

flags
= "flags" flag { flag } [ "/" flag { flag } ] .

with
= "with" | "and" .

icmp
= "icmp-type" icmp-type [ "code" decnumber ] .

return-code = "("icmp-code")" .

keep
= "keep" "state" | "keep" "frags" .

nummask
= host-name [ "/" decnumber ] .

host-name = ipaddr | hostname | "any" .

ipaddr
= host-num "." host-num "." host-num "." host-num .

host-num = digit [ digit [ digit ] ] .

port-num = service-name | decnumber .

withopt = [ "not" | "no" ] opttype [ withopt ] .

opttype = "ipopts" | "short" | "frag" | "opt" ipopts  .

optname
= ipopts [ "," optname ] .

ipopts  = optlist | "sec-class"  .

icmp-type = "unreach" | "echo" | "echorep" | "squench" | "redir" |


    "timex" | "paramprob" | "timest" | "timestrep" | "inforeq" |


    "inforep" | "maskreq" | "maskrep"  | decnumber .

icmp-code = decumber | "net-unr" | "host-unr" | "proto-unr" | "port-unr" |


    "needfrag" | "srcfail" | "net-unk" | "host-unk" | "isolate" |


    "net-prohib" | "host-prohib" | "net-tos" | "host-tos" .

optlist
= "nop" | "rr" | "zsu" | "mtup" | "mtur" | "encode" | "ts" |


  "tr" | "sec" | "lsrr" | "e-sec" | "cipso" | "satid" | "ssrr" |


  "addext" | "visa" | "imitd" | "eip" | "finn" .

hexnumber = "0" "x" hexstring .

hexstring = hexdigit [ hexstring ] .

decnumber = digit [ decnumber ] .

compare = "=" | "!=" | "<" | ">" | "<=" | ">=" | "eq" | "ne" | "lt" |


  "gt" | "le" | "ge" .

range
= "<>" | "><" .

hexdigit = digit | "a" | "b" | "c" | "d" | "e" | "f" .

digit
= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" .

flag
= "F" | "S" | "R" | "P" | "A" | "U" .

This syntax is somewhat simplified for readability, some combinations that match this grammar are disallowed by the software because they do not make sense (such as tcp 'flags' for non-TCP packets).


2.2 Description of the Filter Rules
The "briefest" valid rules are (currently) no-ops and are of the form:

       block in all

       pass in all

       log out all

       count in all

Filter rules are checked in order, with the last matching rule determining the fate of the packet (but see the 'quick' option, below).

Filters are installed by default at the end of the router's filter lists, pre-pending the rule with '@n' will cause it to be inserted as the n'th entry in the current list. This is especially useful when modifying and testing active filter rule sets. 


ACTIONS

The 'action' indicates what to do with the packet if it matches the rest of the filter rule. Each rule MUST have an action. The following actions are recognized:

block
indicates that the packet should be flagged to be dropped. In response to blocking a packet, the filter may be instructed to send a reply packet, either an ICMP packet ('return-icmp'), an ICMP packet masquerading as being from the original packet's destination ('return-icmp-as-dest'), or a TCP "reset" ('return-rst').  An ICMP packet may be generated in response to any IP packet, and its type may optionally be specified, but a TCP reset may only be used with a rule which is being applied to TCP packets.  When using 'return-icmp' or 'return-icmp-as-dest', it is possible to specify the actual unreachable 'type'.  That is, whether it is a network unreachable, port unreachable or even administratively prohibited. This is done by enclosing the ICMP code associated with it in parenthesis directly following 'return-icmp' or 'return-icmp-as-dest' as follows:

        block return-icmp(11) ...

Would return a Type-Of-Service (TOS) ICMP unreachable error.

pass
will flag the packet to be let through the filter.  

Log
causes the packet to be logged (as described in the LOGGING section below) and has no effect on whether the packet will be allowed through the filter.

count

causes the packet to be included in the accounting statistics kept by the filter, and has no effect on whether the packet will be allowed through the filter. These statistics are viewable with 'ipf view' Telnet command.

skip <n>

causes the filter to skip over the next 'n' filter rules.  If a rule is inserted or deleted inside the region being skipped over, then the value of 'n' is adjusted appropriately.

The next word must be either 'in' or 'out'.  Each packet moving through the kernel is either inbound (just been received on an interface, and moving towards the router's protocol processing) or outbound (transmitted or forwarded by the stack, and on its way to an interface). There is a requirement that each filter rule explicitly state which side of the I/O it is to be used on.


OPTIONS

The list of options is brief, and all are indeed optional. Where options are used, they must be present in the order shown here. These are the currently supported options:

log

indicates that, should this be the last matching rule, the packet header will be written to the filter log (as described in the LOGGING section below).

quick

allows "short-cut" rules in order to speed up the filter or override later rules.  If a packet matches a filter rule which is marked as 'quick', this rule will be the last rule checked, allowing a "short-circuit" path to avoid processing later rules for this packet. The current status of the packet (after any effects of the current rule) will determine whether it is passed or blocked.

If this option is missing, the rule is taken to be a "fall-through" rule, meaning that the result of the match (block/pass) is saved and that processing will continue to see if there are any more matches.

on

allows an interface name to be incorporated into the matching procedure. If this option is used, the rule will only match if the packet is going through that interface in the specified direction (in/out). If this option is absent, the rule is to be applied to a packet regardless of the interface it is present on (i.e. on all interfaces). Filter rule sets are common to all interfaces, rather than having a filter list for each interface.

This option is especially useful for simple IP-spoofing protection: packets should only be allowed to pass inbound on the interface from which the specified source address would be expected, others may be logged and/or dropped.

dup-to

causes the packet to be copied, and the duplicate packet to be sent outbound on the specified interface, optionally with the destination IP address changed to that specified. This is useful for off-host logging, using a network sniffer.

MATCHING PARAMETERS

The keywords described in this section are used to describe attributes of the packet to be used when determining whether rules match or don't match. The following general-purpose attributes are provided for matching, and must be used in this order:

tos

packets with different Type-Of-Service values can be filtered. Individual service levels or combinations can be filtered upon.  The value for the TOS mask can either be represented as a hex number or a decimal integer value.

ttl

packets may also be selected by their Time-To-Live value.  The value given in the filter rule must exactly match that in the packet for a match to occur. This value can only be given as a decimal integer value.

proto

allows a specific protocol to be matched against.  All protocol names found in \fB/etc/protocols\fP are recognized and may be used. However, the protocol may also be given as a DECIMAL number, allowing for rules to match your own protocols, or new ones which would out-date any attempted listing.

The special protocol keyword 'tcp/udp' may be used to match either a TCP or a UDP packet, and has been added as a convenience to save duplication of otherwise-identical rules.

The 'from' and 'to' keywords are used to match against IP addresses (and optionally port numbers). Rules must specify BOTH source and destination parameters.

IP addresses may be specified in one of two ways: as a numerical address'/'mask, or as a hostname 'mask' netmask.  The hostname may either be a valid hostname, from either the hosts file or DNS (depending on your configuration and library) or of the dotted numeric form.  There is no special designation for networks but network names are recognized.  Note that having your filter rules depend on DNS results can introduce an avenue of attack, and is discouraged.

There is a special case for the hostname 'any' which is taken to be 0.0.0.0/0 (see below for mask syntax) and matches all IP addresses. Only the presence of "any" has an implied mask, in all other situations, a hostname MUST be accompanied by a mask.  It is possible to give "any" a hostmask, but in the context of this language, it is non-sensical.

The numerical format "x'/'y" indicates that a mask of y consecutive 1 bits set is generated, starting with the MSB, so a y value of 16 would give 0xffff0000. The symbolic "x 'mask' y" indicates that the mask y is in dotted IP notation or a hexadecimal number of the form 0x12345678.  Note that all the bits of the IP address indicated by the bit-mask must match the address on the packet exactly; there isn't currently a way to invert the sense of the match, or to match ranges of IP addresses which do not express themselves easily as bit-masks. (anthropomorphization; it's not just for breakfast anymore).

If a 'port' match is included, for either or both of source and destination, then it is only applied to TCP and UDP packets. 


If there is no 'proto' match parameter, packets from both protocols are compared. This is equivalent to "proto tcp/udp".  When composing 'port' comparisons, either the service name or an integer port number may be used. Port comparisons may be done in a number of forms, with a number of comparison operators, or port ranges may be specified. When the port appears as part of the 'from' object, it matches the source port number, when it appears as part of the 'to' object, it matches the destination port number. See the examples for more information.

The 'all' keyword is essentially a synonym for "from any to any" with no other match parameters.

Following the source and destination matching parameters, the following additional parameters may be used:

with

is used to match irregular attributes that some packets may have associated with them.  To match the presence of IP options in general, use 'with ipopts'. To match packets that are too short to contain a complete header, use 'with short'. To match fragmented packets, use 'with frag'.  For more specific filtering on IP options, individual options can be listed.

Before any parameter used after the 'with' keyword, the word 'not' or 'no' may be inserted to cause the filter rule to only match if the option(s) is not present.

Multiple consecutive 'with' clauses are allowed.  Alternatively, the keyword 'and' may be used in place of 'with', this is provided purely to make the rules more readable ("with ... and ..."). When multiple clauses are listed, all those must match to cause a match of the rule.


flags

is only effective for TCP filtering.  Each of the letters possible represents one of the possible flags that can be set in the TCP header.  The association is as follows:

        F - FIN

        S - SYN

        R - RST

        P - PUSH

        A - ACK

        U - URG

The various flag symbols may be used in combination, so that "SA" would represent a SYN-ACK combination present in a packet.  There is nothing preventing the specification of combinations, such as "SFR", that would not normally be generated by law-abiding TCP implementations.  However, to guard against weird aberrations, it is necessary to state which flags you are filtering against.  To allow this, it is possible to set a mask indicating which TCP flags you wish to compare (i.e., those you deem significant).  This is done by appending "/<flags>" to the set of TCP flags you wish to match against, e.g.:


... flags S




# becomes "flags S/AUPRFS" and will match




# packets with ONLY the SYN flag set.


... flags SA




# becomes "flags SA/AUPRFS" and will match any




# packet with only the SYN and ACK flags set.


... flags S/SA




# will match any packet with just the SYN flag set




# out of the SYN-ACK pair; the common "establish"




# keyword action.  "S/SA" will NOT match a packet




# with BOTH SYN and ACK set, but WILL match "SFP".

icmp-type

is only effective when used with 'proto icmp' and must NOT be used in conjunction with 'flags'.  There are a number of types, which can be referred to by an abbreviation recognized by this language, or the numbers with which they are associated can be used.  The most important from a security point of view is the ICMP redirect.
KEEP HISTORY

The second last parameter that can be set for a filter rule is whether or not to record historical information for that packet, and what sort to keep. The following information can be kept:

state

keeps information about the flow of a communication session. State can be kept for TCP, UDP, and ICMP packets.

frags

keeps information on fragmented packets, to be applied to later fragments.

allowing packets which match these to flow straight through, rather than going through the access control list.
GROUPS
The last pair of parameter control filter rule "grouping".  By default, all filter rules are placed in group 0 if no other group is specified.  To add a rule to a non-default group, the group must first be started by creating a group 'head'.  If a packet matches a rule which is the 'head' of a group, the filter processing then switches to the group, using that rule as the default for the group.  
Note: If 'quick' is used with a 'head' rule, rule processing isn't stopped until it has returned from processing the group.

A rule may be both the head for a new group and a member of a non-default group ('head' and 'group' may be used together in a rule).

head <n>

indicates that a new group (number n) should be created.

group <n>

indicates that the rule should be put in group (number n) rather than group 0.
LOGGING

When a packet is logged, with either the 'log' action or option, the headers of the packet are written to the embedded packet logging memory. Immediately following the 'log' keyword, the following qualifiers may be used (in order):

body

indicates that the first 128 bytes of the packet contents will be logged after the headers. 

first

If log is being used in conjunction with a "keep" option, it is recommended that this option is also applied so that only the triggering packet is logged and not every packet which thereafter matches state information.

EXAMPLES

The 'quick' option is good for rules such as:

block in quick from any to any with ipopts

which will match any packet with a non-standard header length (IP options present) and abort further processing of later rules, recording a match and also that the packet should be blocked.

The "fall-through" rule parsing allows for effects such as this:

        block in from any to any port < 6000

        pass in from any to any port >= 6000

        block in from any to any port > 6003

which sets up the range 6000-6003 as being permitted and all others being denied.  Note that the effect of the first rule is overridden by subsequent rules.  Another (easier) way to do the same is:

        block in from any to any port 6000 <> 6003

        pass in from any to any port 5999 >< 6004

Note that both the "block" and "pass" are needed here to effect a result as a failed match on the "block" action does not imply a pass, only that the rule hasn't taken effect.  To then allow ports < 1024, a rule such as:

        pass in quick from any to any port < 1024

would be needed before the first block.  To create a new group for processing all inbould packets on 'lan/isp/dialin', with the default being to block all inbound packets, we would do something like:

       block in all

       block in quick on lan all head 100

       block in quick on isp all head 200

       block in quick on dialin all head 300

and to then allow ICMP packets in on 'lan', only, we would do:

       pass in proto icmp all group 100

Note that because only inbound packets on 'lan' are used processed by group 100, there is no need to re-specify the interface name.  Likewise, we could further breakup processing of TCP, etc, as follows:

       block in proto tcp all head 110 group 100

       pass in from any to any port = 23 group 110

and so on.  The last line, if written without the groups would be: 
       pass in on lan proto tcp from any to any port = telnet

Note, that if we wanted to say "port = telnet", "proto tcp" would need to be specified as the parser interprets each rule on its own and qualifies all service/port names with the protocol specified.

3. Operations of the IP Filter

All IP Filter's operations are made through the Telnet of the ISDN router. Three major Telenet commands are supported:

ipf 

alters packet filtering lists for IP packet input and output
ipf view
retrieves and displays the accumulated statistics which have been accumulated over time as the router has put packets through the filter.

log
displays the filter packets logged, the PPP/MP logs and the ISDN D-channel protocols in a human readable format.
3.1 The ipf Command
Usages:

ipf [-ADEInorstVzZ] [-l block|pass|nomatch] [-F i|o|a|s|S] [-f rule]

 -A 
Set the list to make changes to the active list (default).

 -D     Disable the filter (if enabled).  

 -E     Enable the filter (if disabled). 
 -I     Set the list to make changes to the inactive list.

 -n     This flag (no-change) prevents 'ipf' from actually doing anything which would alter the currently running configuration.

 -o     Force rules by default to be added/deleted to/from the output list, rather than the (default) input list.

 -r     Remove matching filter rules rather than add them to the internal lists

 -s     Swap the active filter list in use to be the "other" one.

 -t     Display to the end. (disable the 'more' function of the Telnet display)
 -V     Show version IP Filter information. Information about its current state will also be displayed (whether logging is active, default filtering, etc).

 -z     For each rule in the input file, reset the statistics for it to zero and display the statistics prior to them being zero'd.

-Z     
Zero global statistics held in the kernel for filtering only (this doesn't affect fragment or state statistics).

-l     
Toggle default logging of packets.  Valid arguments to this option are 'pass', 'block' and 'nomatch'. Any other arguments will cause this logging function to be disabled. When a valid option is set, any packet that exits filtering and matches the set category is logged.  This is most useful for causing all packets that don't match any of the loaded rules to be logged.

-F     
This option specifies which filter list to flush.  The parameter should either be "i" (input), "o" (output) or "a" (remove all filter rules). 
-F  s
Remove state information about any non-fully established connections.

-F  S
Delete the entire state table. A fully established connection will show up in 'ipf view -s' output as 4/4, with deviations either way indicating it is not fully established any more.

-f     
key-in the filter rule (type 'ipf rule ?' for details)

3.2 The ipf view Command

The role of 'ipf view' is to display current kernel statistics gathered as a result of applying the filters in place (if any) to packets going in and out of the router.  This is the default operation when no command line parameters are present.

When supplied with either '-i' or '-o', it will retrieve and display the appropriate list of filter rules currently installed and in use by the router.
Usage:

 ipf view [-afhinostz] [-g <group>] [-l <line>]

-a 

Display the accounting filter list and show bytes counted against each rule.

-f      Show fragment state information (statistics) and held state information, if any.

-h      Show per-rule the number of times each one scores a "hit".  For use in combination with '-i'.
-i      Display the filter list used for the input side of the router IP processing.

-n      For the inactive filter set

-o      Display the filter list used for the output side of the router IP processing.

-s      Show packet/flow state information (statistics) and held state information  if any is present.

-t      Display to the end. (Disable the 'more' function of the Telnet display)
-z     
Clear the statistics after this command

-g <group>      specify the group number

-l <line>       specify the rule's line number within a list/group

3.3 The log Command

There are four ring buffers used in the ISDN router to log four types of data:

Wan
Log

Include ISDN D-channel protocol as well as the PPP protocols

Call Log 
Record the establishment of a connection

Filter Log
Record packets according to the filter rule as well as the 'ipf -l' command.

State Log
Record the IP packet state according to the 'keep state' filter rule
The Wan Log and Call Log are always enabled. Once the buffer is about to overflow, the oldest data segment recorded will be overwritten.

Usage:

log [-cfhipstwx?][-F a|c|f|s|w]

-c  
Display the call log

-f      Display the filter log

-F      Flush the log buffers
  a             flush all logs

  c             flush the call log

  f             flush the filter log

  s             flush the state log

  w             flush the wan (ISDN and PPP) log

-h      For this usage help

-i      Display the ISDN D-channel part of the wan log
-p      Display the PPP/MP part of the wan log
-s      Display the state log

-t      Display to the end. (Disable the 'more' function of the Telnet display) 
-w      Display the wan (ISDN and PPP) log

-x      For packet body hex dump, if any

